# Transforming the parametric equations of a curve into an implicit equation in Cartesian coordinates

I have a parametric curve defined by

x = 2 Sin[t] + Sin[3 t]
y = 2 Cos[t] - Cos[3 t]


How can I eliminate t so get a single implicit equation in x and y?

• Non-implicit solution: Solve[{x == 2 Sin[t] + Sin[3 t], y == 2 Cos[t] - Cos[3 t]}, {t, y}]. – T.F Jul 24 '17 at 5:20
• have you tried the code? When you solve for y,t you automatically eliminate t... Hence the equation is expressed in terms of x and y. Next time try running the code... – T.F Jul 24 '17 at 6:18
• @T.F: Sorry. y=f(x) is not what I need. I need f(x, x^2, x^3, ..., y, y^2, y^3,...)=0. :-) – Kim Jong Un Jul 24 '17 at 6:49

You should first rationalize the trig functions and the use Eliminate:

Clear[x, y]

Eliminate[
TrigExpand[{x == 2 Sin[t] + Sin[3 t], y == 2 Cos[t] - Cos[3 t]} /.
t -> 2 ArcTan[u]], u]

(*
==> x^6 + x^4 (17 + 3 y^2) + x^2 (63 - 94 y^2 + 3 y^4) ==
81 - 63 y^2 - 17 y^4 - y^6
*)


The replacement rule introduces the rationalization with variable u.